

LAB 05

梁峻滔 PB19051175

1. 算法思想

本实验只需要我们实现两个程序，一个程序要求循环输出“ICS2020”字符串，一个程序要求从键盘输入一个字符，判断该字符是否是十进制数字后返回循环输出“ICS2020”的程序。

第一个程序，使用.STRINGZ 伪指令预先在内存中存放“ICS2020 ”字符串(这里加多了一个空格符，这样就不必添加额外的单独用以输出空格的指令)，然后通过一个指针寄存器和 R0 读入字符和使用 TRAP x21 输出。每次输出完一个“ICS2020 ”字符串后通过 JSR 调用一个计数程序，以减缓字符串输出的频率。

第二个程序，预先在内存中存好‘0’和‘9’的 ASCII 码的相反数，每次输入一个字符，都与这两个 ASCII 码通过 ADD 比较，若输入字符的 ASCII 码不小于 ASCII(0)且不大于 ASCII(9)，则该字符就是一个十进制数字，否则就不是。同样，使用.STRINGZ 伪指令在内存中存放“ is not a decimal digit.”和“ is a decimal digit.”字符串，判断输入字符是否十进制数字后再输出一该字符，紧接着输出相应的判断字符串即可，随后使用 RTI 返回第一个程序。

2. 代码实现

(1) Starter Code

```

15      .ORIG x800
16      ; (1) Initialize interrupt vector table.
17      LD R0, VEC
18      LD R1, ISR
19      STR R1, R0, #0
20
21      ; (2) Set bit 14 of KBSR.
22      LDI R0, KBSR
23      LD R1, MASK
24      NOT R1, R1
25      AND R0, R0, R1
26      NOT R1, R1
27      ADD R0, R0, R1
28      STI R0, KBSR
29
30      ; (3) Set up system stack to enter user space.
31      LD R0, PSR
32      ADD R6, R6, #-1
33      STR R0, R6, #0
34      LD R0, PC
35      ADD R6, R6, #-1
36      STR R0, R6, #0
37      ; Enter user space.
38      RTI
39
40 VEC   .FILL x0180
41 ISR   .FILL x1000
42 KBSR  .FILL xFE00
43 MASK  .FILL x4000
44 PSR    .FILL x8002
45 PC    .FILL x3000
46      .END

```

(2) 补充部分

```

48      .ORIG x3000
49      ; *** Begin user program code here ***
50 LDSTR LEA    R1, STRING
51 Print LDR    R0, R1, #0      ; 输出 "ICS2020 "
52      BRz    CALL            ; 输出完一个字符串后调用计数程序
53      TRAP   x21
54      ADD    R1, R1, #1
55      BRnzp  Print
56 CALL    JSR    DELAY
57      BRnzp  LDSTR
58 DELAY   ST     R1, SaveR1
59      LD     R1, COUNT
60 REP     ADD    R1, R1, #-1
61      BRp    REP
62      LD     R1, SaveR1
63      RET
64 COUNT  .FILL x7FFF
65 STRING .STRINGZ "ICS2020 "
66 SaveR1 .BLKW #1
67      ; *** End user program code here ***
68      .END

```

```

70      .ORIG x1000
71      ; *** Begin interrupt service routine code here ***
72      ST      R0,SaveR0
73      ST      R1,SaveR1_2
74      ST      R2,SaveR2
75      ST      R3,SaveR3
76      TRAP    x20      ;从键盘读入一个字符到R0
77      TRAP    x21      ;Echo
78      LD      R2,ASCII0 ;输入的字符与'0'比较
79      ADD     R2,R0,R2
80      BRn     FALSE
81      LD      R2,ASCII9 ;输入的字符与'9'比较
82      ADD     R2,R0,R2
83      BRp     FALSE
84  TRUE  ADD     R3,R0,#0 ;由于要输出换行符,先将该字符存到R3
85          LD     R0,NEWLINE
86          TRAP   x21
87          ADD    R0,R3,#0 ;将该十进制数字字符放回R0后输出
88          TRAP   x21
89          LEA    R1,POS ;紧接着输出肯定的字符串
90  Print1 LDR     R0,R1,#0
91          BRz    FINISH
92          TRAP   x21
93          ADD    R1,R1,#1
94          BRnzp Print1
95  FALSE ADD     R3,R0,#0 ;由于要输出换行符,先将该字符存到R3
96          LD     R0,NEWLINE
97          TRAP   x21
98          ADD    R0,R3,#0 ;将该十进制数字字符放回R0后输出
99          TRAP   x21
100         LEA    R1,NEG ;紧接着输出否定的字符串
101  Print2 LDR     R0,R1,#0
102         BRz    FINISH
103         TRAP   x21
104         ADD    R1,R1,#1

```

```

105         BRnzp Print2
106  FINISH LD      R0,NEWLINE
107         TRAP   x21
108         LD     R0,SaveR0
109         LD     R1,SaveR1_2
110         LD     R2,SaveR2
111         LD     R3,SaveR3
112         RTI
113  SaveR0 .BLKW #1
114  SaveR1_2 .BLKW #1
115  SaveR2 .BLKW #1
116  SaveR3 .BLKW #1
117  NEG     .STRINGZ " is not a decimal digit."
118  POS     .STRINGZ " is a decimal digit."
119  ASCII0 .FILL xFFD0 ;-x0030 '0'
120  ASCII9 .FILL xFFC7 ;-x0039 '9'
121  NEWLINE .FILL x000A
122      ; *** End interrupt service routine code here ***
123      .END

```

3. 测试

```
ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020
ICS2020 ICS2020 ICS2020 a
a is not a decimal digit.
ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020
ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 3
3 is a decimal digit.
ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020
ICS2020 ICS2020 6
6 is a decimal digit.
ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 f
f is not a decimal digit.
ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 %
% is not a decimal digit.
ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 #
# is not a decimal digit.
ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 &
& is not a decimal digit.
ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 d
d is not a decimal digit.
ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020
ICS2020 ICS2020 ICS2020 ICS2020 2
2 is a decimal digit.
ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020
ICS2020 0
0 is a decimal digit.
ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 ICS2020 □
```

对于以上随机输入的字符，程序都能正确判断并反应。